

Формальные методы в программной инженерии

Раздел 1.5: Практическое занятие

Чушкин Михаил Сергеевич
chushkinm@rambler.ru

Материалы

- Лекции
- Задачи для самостоятельного решения
- Дисциплинарная ведомость

Решатели

- Z3 online
 - <http://rise4fun.com/z3>
- Z3 offline
 - <https://z3.codeplex.com/releases>
- Yices
 - <http://yices.csl.sri.com/>

Задача

Показать, что для любой ненулевой целочисленной матрицы A третьего порядка выполняется тождество $AE = A$, где E – единичная матрица.

Матрица третьего порядка

$$A = \begin{pmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{pmatrix}$$

- Двумерный массив
(declare-fun A () (array Int (array Int Int)))
- Отображение (функция)
(declare-fun A (Int Int) Int)

Решение задачи

```
(set-logic UFNIA)
```

```
(declare-fun a (Int Int) Int)
```

Условия для функции a

$$a : \text{Int} \times \text{Int} \rightarrow \text{Int}$$

$$\exists i, j \in \text{Int} \quad a(i, j) \neq 0$$

(exists ((i Int) (j Int)) (not (= (a i j) 0)))

$$\forall i, j \in \text{Int} \quad (i < 1 \vee i > 3 \vee j < 1 \vee j > 3) \Rightarrow a(i, j) = 0$$

(forall ((i Int) (j Int))

(=>

(or (<= i 0) (>= i 4) (<= j 0) (>= j 4))

(= (a i j) 0)))

Решение задачи

```
(set-logic UFNIA)

(declare-fun a (Int Int) Int)

(assert (exists ((i Int) (j Int)) (not (= (a i j) 0))))

(assert
  (forall ((i Int) (j Int))
    (=>
      (or (<= i 0) (>= i 4) (<= j 0) (>= j 4))
      (= (a i j) 0)
    )
  )
)
```


Единичная матрица

$$E = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(define-fun e ((i Int) (j Int)) Int
  (ite (and (= i j) (> i 0) (< i 4)) 1 0)
)
```

Произведение двух матриц

$$(AB)_{ij} = \sum_{k=1}^n A_{ik} B_{kj}$$

```
(define-fun ae ((i Int) (j Int)) Int
  (ite
    (and (> i 0) (< i 4) (> j 0) (< j 4))
    (+
      (* (a i 1) (e 1 j))
      (* (a i 2) (e 2 j))
      (* (a i 3) (e 3 j))
    )
    0
  )
)
```

Утверждение

$$\forall A \quad AE = A$$

$$\neg(\forall A \quad AE = A)$$

$$\exists A \quad AE \neq A$$

$$\neg(\forall i, j \in Int \quad ae(i, j) = a(i, j))$$

Решение задачи

```
(assert
  (not
    (forall ((i Int) (j Int))
      (= (ae i j) (a i j))
    )
  )
)
```

```
(check-sat)
```

```
(get-model)
```

Задача

Для любых двух ненулевых целочисленных матриц третьего порядка A и B показать, что выполняется тождество $tr(AB) = tr(BA)$, где tr – след матрицы.

$$tr(A) = \sum_{i=1}^n A_{ii}$$

- 1) Определить функции a и b
- 2) Определить функции ab и ba
- 3) Определить функции $tr-ab$ и $tr-ba$
- 4) $tr-ab \neq tr-ba$